********Table O.1 - Mexico********

eststo clear

*Load data and create differences
cd "$reploc/data/created_data"
use mex_full, clear

sort pid_link mxfls
drop if mxfls==.
keep if panel==1

*Generate age dummies, and rename to avoid confusion
tab imp_age, gen(impagedum)

foreach var of varlist impagedum* {
local nam1=substr("`var'",10,.)
rename `var' impage`nam1'dum
}

keep rbuc* mxfls pid_link inflation imp_age panel birthyr grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co? birthstate yearprov impage*dum

local varsdumsage
foreach var of varlist impage*dum {
local varsdumsage `varsdumsage' `var'
}

reshape wide rbuc*  inflation imp_age panel grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co? `varsdumsage', i(pid_link) j(mxfls)


local varlist1 rbuc rbuc_nga inflation grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co2 grbrprovSD_lam0co2 grbrprovAvg_lam0co3 grbrprovSD_lam0co3 `varsdumsage'


foreach var in `varlist1'  {
gen `var'dif=.
replace `var'dif=`var'3-`var'2
}

*Generate cohort by birthyear variables to cluster on
gen coh_prov_clust=yearprov

*Regressions
local x rbuc_nga
local rhsvari oldmexonly
local rhsvaruse _lam0co3dif
local usei grbr

drop if `x'dif==.

*Column 1
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)

*Column 2
eststo:  reg `x'dif `usei'provSD`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)

*Column 3
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse' `usei'provSD`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)


cd "$reploc/results"	
esttab using TableO1_Mexico.tex, star(* 0.1 ** 0.05 *** 0.01) b(2) se(2) replace
